NAG Library Routine Document

F08YLF (DTGSNA)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08YLF (DTGSNA) estimates condition numbers for specified eigenvalues and/or eigenvectors of a matrix pair in generalized real Schur form.

2 Specification

```
SUBROUTINE F08YLF (JOB, HOWMNY, SELECT, N, A, LDA, B, LDB, VL, LDVL, VR,      &
                   LDVR, S, DIF, MM, M, WORK, LWORK, IWORK, INFO)

INTEGER            N, LDA, LDB, LDVL, LDVR, MM, M, LWORK, IWORK(*), INFO
REAL (KIND=nag_wp) A(LDA,*), B(LDB,*), VL(LDVL,*), VR(LDVR,*), S(*),          &
                   DIF(*), WORK(max(1,LWORK))
LOGICAL            SELECT(*)
CHARACTER(1)       JOB, HOWMNY
```

The routine may be called by its LAPACK name ***dtgsna***.

3 Description

F08YLF (DTGSNA) estimates condition numbers for specified eigenvalues and/or right eigenvectors of an n by n matrix pair (S,T) in real generalized Schur form. The routine actually returns estimates of the reciprocals of the condition numbers in order to avoid possible overflow.

The pair (S,T) are in real generalized Schur form if S is block upper triangular with 1 by 1 and 2 by 2 diagonal blocks and T is upper triangular as returned, for example, by F08XAF (DGGES) or F08XBF (DGGESX), or F08XEF (DHGEQZ) with JOB = 'S'. The diagonal elements, or blocks, define the generalized eigenvalues (α_i,β_i), for $i=1,2,\ldots,n$, of the pair (S,T) and the eigenvalues are given by

$$\lambda_i=\alpha_i/\beta_i,$$

so that

$$\beta_i S x_i=\alpha_i T x_i \qquad \text{or} \qquad S x_i=\lambda_i T x_i,$$

where x_i is the corresponding (right) eigenvector.

If S and T are the result of a generalized Schur factorization of a matrix pair (A,B)

$$A=QSZ^{\mathrm{T}}, \qquad B=QTZ^{\mathrm{T}}$$

then the eigenvalues and condition numbers of the pair (S,T) are the same as those of the pair (A,B).

Let $(\alpha,\beta)\neq(0,0)$ be a simple generalized eigenvalue of (A,B). Then the reciprocal of the condition number of the eigenvalue $\lambda=\alpha/\beta$ is defined as

$$s(\lambda)=\frac{\left(\left|y^{\mathrm{T}}Ax\right|^2+\left|y^{\mathrm{T}}Bx\right|^2\right)^{1/2}}{\left(\|x\|_2\|y\|_2\right)},$$

where x and y are the right and left eigenvectors of (A,B) corresponding to λ. If both α and β are zero, then (A,B) is singular and $s(\lambda)=-1$ is returned.

The definition of the reciprocal of the estimated condition number of the right eigenvector x and the left eigenvector y corresponding to the simple eigenvalue λ depends upon whether λ is a real eigenvalue, or one of a complex conjugate pair.

If the eigenvalue λ is real and U and V are orthogonal transformations such that

$$U^{\mathrm{T}}(A,B)V = (S,T) = \begin{pmatrix} \alpha & * \\ 0 & S_{22} \end{pmatrix} \begin{pmatrix} \beta & * \\ 0 & T_{22} \end{pmatrix},$$

where S_{22} and T_{22} are $(n-1)$ by $(n-1)$ matrices, then the reciprocal condition number is given by

$$\mathrm{Dif}(x) \equiv \mathrm{Dif}(y) = \mathrm{Dif}((\alpha,\beta),(S_{22},T_{22})) = \sigma_{\min}(Z),$$

where $\sigma_{\min}(Z)$ denotes the smallest singular value of the $2(n-1)$ by $2(n-1)$ matrix

$$Z = \begin{pmatrix} \alpha \otimes I & -1 \otimes S_{22} \\ \beta \otimes I & -1 \otimes T_{22} \end{pmatrix}$$

and $\otimes$ is the Kronecker product.

If λ is part of a complex conjugate pair and U and V are orthogonal transformations such that

$$U^{\mathrm{T}}(A,B)V = (S,T) = \begin{pmatrix} S_{11} & * \\ 0 & S_{22} \end{pmatrix} \begin{pmatrix} T_{11} & * \\ 0 & T_{22} \end{pmatrix},$$

where S_{11} and T_{11} are two by two matrices, S_{22} and T_{22} are $(n-2)$ by $(n-2)$ matrices, and (S_{11},T_{11}) corresponds to the complex conjugate eigenvalue pair λ, $\bar{\lambda}$, then there exist unitary matrices U_1 and V_1 such that

$$U_1^H S_{11} V_1 = \begin{pmatrix} s_{11} & s_{12} \\ 0 & s_{22} \end{pmatrix} \quad \text{and} \quad U_1^H T_{11} V_1 = \begin{pmatrix} t_{11} & t_{12} \\ 0 & t_{22} \end{pmatrix}.$$

The eigenvalues are given by $\lambda = s_{11}/t_{11}$ and $\bar{\lambda} = s_{22}/t_{22}$. Then the Frobenius norm-based, estimated reciprocal condition number is bounded by

$$\mathrm{Dif}(x) \equiv \mathrm{Dif}(y) \leq \min(d_1, \max(1, |\mathrm{Re}(s_{11})/\mathrm{Re}(s_{22})|), d_2)$$

where $\mathrm{Re}(z)$ denotes the real part of z, $d_1 = \mathrm{Dif}((s_{11},t_{11}),(s_{22},t_{22})) = \sigma_{\min}(Z_1)$, Z_1 is the complex two by two matrix

$$Z_1 = \begin{pmatrix} s_{11} & -s_{22} \\ t_{11} & -t_{22} \end{pmatrix},$$

and d_2 is an upper bound on $\mathrm{Dif}((S_{11},T_{11}),(S_{22},T_{22}))$; i.e., an upper bound on $\sigma_{\min}(Z_2)$, where Z_2 is the $(2n-2)$ by $(2n-2)$ matrix

$$Z_2 = \begin{pmatrix} S_{11}^T \otimes I & -I \otimes S_{22} \\ T_{11}^T \otimes I & -I \otimes T_{22} \end{pmatrix}.$$

See Sections 2.4.8 and 4.11 of Anderson *et al.* (1999) and Kågström and Poromaa (1996) for further details and information.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

Kågström B and Poromaa P (1996) LAPACK-style algorithms and software for solving the generalized Sylvester equation and estimating the separation between regular matrix pairs *ACM Trans. Math. Software* **22** 78–103

5 Parameters

1: JOB – CHARACTER(1) *Input*

On entry: indicates whether condition numbers are required for eigenvalues and/or eigenvectors.

JOB = 'E'

Condition numbers for eigenvalues only are computed.

JOB = 'V'
Condition numbers for eigenvectors only are computed.

JOB = 'B'
Condition numbers for both eigenvalues and eigenvectors are computed.

Constraint: JOB = 'E', 'V' or 'B'.

2: HOWMNY – CHARACTER(1) *Input*

On entry: indicates how many condition numbers are to be computed.

HOWMNY = 'A'
Condition numbers for all eigenpairs are computed.

HOWMNY = 'S'
Condition numbers for selected eigenpairs (as specified by SELECT) are computed.

Constraint: HOWMNY = 'A' or 'S'.

3: SELECT($*$) – LOGICAL array *Input*

Note: the dimension of the array SELECT must be at least $\max(1, \mathrm{N})$ if HOWMNY = 'S', and at least 1 otherwise.

On entry: specifies the eigenpairs for which condition numbers are to be computed if HOWMNY = 'S'. To select condition numbers for the eigenpair corresponding to the real eigenvalue λ_j, SELECT(j) must be set .TRUE.. To select condition numbers corresponding to a complex conjugate pair of eigenvalues λ_j and λ_{j+1}, SELECT(j) and/or SELECT($j+1$) must be set to .TRUE..

If HOWMNY = 'A', SELECT is not referenced.

4: N – INTEGER *Input*

On entry: n, the order of the matrix pair (S, T).

Constraint: $\mathrm{N} \geq 0$.

5: A(LDA,$*$) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the upper quasi-triangular matrix S.

6: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08YLF (DTGSNA) is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{N})$.

7: B(LDB,$*$) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{N})$.

On entry: the upper triangular matrix T.

8: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F08YLF (DTGSNA) is called.

Constraint: $\mathrm{LDB} \geq \max(1, \mathrm{N})$.

9: VL(LDVL,$*$) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array VL must be at least $\max(1, \text{MM})$ if JOB = 'E' or 'B', and at least 1 otherwise.

On entry: if JOB = 'E' or 'B', VL must contain left eigenvectors of (S, T), corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors must be stored in consecutive columns of VL, as returned by F08WAF (DGGEV) or F08YKF (DTGEVC).

If JOB = 'V', VL is not referenced.

10: LDVL – INTEGER *Input*

On entry: the first dimension of the array VL as declared in the (sub)program from which F08YLF (DTGSNA) is called.

Constraints:

if JOB = 'E' or 'B', LDVL $\geq \max(1, \text{N})$;
otherwise LDVL ≥ 1.

11: VR(LDVR,$*$) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array VR must be at least $\max(1, \text{MM})$ if JOB = 'E' or 'B', and at least 1 otherwise.

On entry: if JOB = 'E' or 'B', VR must contain right eigenvectors of (S, T), corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors must be stored in consecutive columns of VR, as returned by F08WAF (DGGEV) or F08YKF (DTGEVC).

If JOB = 'V', VR is not referenced.

12: LDVR – INTEGER *Input*

On entry: the first dimension of the array VR as declared in the (sub)program from which F08YLF (DTGSNA) is called.

Constraints:

if JOB = 'E' or 'B', LDVR $\geq \max(1, \text{N})$;
otherwise LDVR ≥ 1.

13: S($*$) – REAL (KIND=nag_wp) array *Output*

Note: the dimension of the array S must be at least $\max(1, \text{MM})$.

On exit: if JOB = 'E' or 'B', the reciprocal condition numbers of the selected eigenvalues, stored in consecutive elements of the array. For a complex conjugate pair of eigenvalues two consecutive elements of S are set to the same value. Thus S(j), DIF(j), and the jth columns of VL and VR all correspond to the same eigenpair (but not in general the jth eigenpair, unless all eigenpairs are selected).

If JOB = 'V', S is not referenced.

14: DIF($*$) – REAL (KIND=nag_wp) array *Output*

Note: the dimension of the array DIF must be at least $\max(1, \text{MM})$.

On exit: if JOB = 'V' or 'B', the estimated reciprocal condition numbers of the selected eigenvectors, stored in consecutive elements of the array. For a complex eigenvector two consecutive elements of DIF are set to the same value. If the eigenvalues cannot be reordered to compute DIF(j), DIF(j) is set to 0; this can only occur when the true value would be very small anyway.

If JOB = 'E', DIF is not referenced.

15: MM – INTEGER *Input*

On entry: the number of elements in the arrays S and DIF.

Constraint: $\mathrm{MM} \geq \mathrm{N}$.

16: M – INTEGER *Output*

On exit: the number of elements of the arrays S and DIF used to store the specified condition numbers; for each selected real eigenvalue one element is used, and for each selected complex conjugate pair of eigenvalues, two elements are used. If HOWMNY = 'A', M is set to N.

17: WORK(max(1, LWORK)) – REAL (KIND=nag_wp) array *Workspace*

On exit: if INFO $= 0$, WORK(1) contains the minimum value of LWORK required for optimal performance.

18: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08YLF (DTGSNA) is called.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the minimum size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Constraints: if LWORK $\neq -1$,

if JOB = 'V' or 'B', LWORK $\geq 2 \times \mathrm{N} \times (\mathrm{N}+2) + 16$;
otherwise LWORK $\geq \max(1, \mathrm{N})$.

19: IWORK($*$) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $(\mathrm{N}+6)$.

If JOB = 'E', IWORK is not referenced.

20: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

None.

8 Further Comments

An approximate asymptotic error bound on the chordal distance between the computed eigenvalue $\tilde{\lambda}$ and the corresponding exact eigenvalue λ is

$$\chi(\tilde{\lambda}, \lambda) \leq \epsilon \|(A, B)\|_F / S(\lambda)$$

where ϵ is the ***machine precision***.

An approximate asymptotic error bound for the right or left computed eigenvectors $\tilde{x}$ or $\tilde{y}$ corresponding to the right and left eigenvectors x and y is given by

$$\theta(\tilde{z}, z) \leq \epsilon \|(A, B)\|_F / \text{Dif}.$$

The complex analogue of this routine is F08YYF (ZTGSNA).

9 Example

This example estimates condition numbers and approximate error estimates for all the eigenvalues and eigenvalues and right eigenvectors of the pair (S, T) given by

$$S = \begin{pmatrix} 4.0 & 1.0 & 1.0 & 2.0 \\ 0 & 3.0 & -1.0 & 1.0 \\ 0 & 1.0 & 3.0 & 1.0 \\ 0 & 0 & 0 & 6.0 \end{pmatrix} \quad \text{and} \quad T = \begin{pmatrix} 2.0 & 1.0 & 1.0 & 3.0 \\ 0 & 1.0 & 0.0 & 1.0 \\ 0 & 0 & 1.0 & 1.0 \\ 0 & 0 & 0 & 2.0 \end{pmatrix}.$$

The eigenvalues and eigenvectors are computed by calling F08YKF (DTGEVC).

9.1 Program Text

```
    Program f08ylfe

!     F08YLF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: dtgevc, dtgsna, f06bnf, f06raf, nag_wp, x02ajf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)                :: eps, snorm, stnrm, tnorm
      Integer                           :: i, info, lda, ldb, ldvl, ldvr,       &
                                           lwork, m, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable   :: a(:,:), b(:,:), dif(:), s(:),        &
                                           vl(:,:), vr(:,:), work(:)
      Integer, Allocatable              :: iwork(:)
      Logical                           :: select(1)
!     .. Executable Statements ..
      Write (nout,*) 'F08YLF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      ldb = n
      ldvl = n
      ldvr = n
      lwork = 2*n*(n+2) + 16
      Allocate (a(lda,n),b(ldb,n),dif(n),s(n),vl(ldvl,n),vr(ldvr,n), &
        work(lwork),iwork(n+6))

!     Read A and B from data file

      Read (nin,*)(a(i,1:n),i=1,n)
      Read (nin,*)(b(i,1:n),i=1,n)

!     Calculate the left and right generalized eigenvectors of the
!     pair (A,B). Note that DTGEVC requires WORK to be of dimension
!     at least 6*n.

!     The NAG name equivalent of dtgevc is f08ykf
      Call dtgevc('Both','All',select,n,a,lda,b,ldb,vl,ldvl,vr,ldvr,n,m,work, &
        info)

      If (info>0) Then
```

```
        Write (nout,99999) info, info + 1
      Else

!       Estimate condition numbers for all the generalized eigenvalues
!       and right eigenvectors of the pair (A,B)

!       The NAG name equivalent of dtgsna is f08ylf
        Call dtgsna('Both','All',select,n,a,lda,b,ldb,vl,ldvl,vr,ldvr,s,dif,n, &
          m,work,lwork,iwork,info)

!       Print condition numbers of eigenvalues and right eigenvectors

        Write (nout,*) 'S'
        Write (nout,99998) s(1:m)
        Write (nout,*)
        Write (nout,*) 'DIF'
        Write (nout,99998) dif(1:m)

!       Calculate approximate error estimates

!       Compute the 1-norms of A and B and then compute
!       SQRT(snorm**2 + tnorm**2)

        eps = x02ajf()
        snorm = f06raf('1-norm',n,n,a,lda,work)
        tnorm = f06raf('1-norm',n,n,b,ldb,work)
        stnrm = f06bnf(snorm,tnorm)
        Write (nout,*)
        Write (nout,*) 'Approximate error estimates for eigenvalues of (A,B)'
        Write (nout,99998)(eps*stnrm/s(i),i=1,m)
        Write (nout,*)
        Write (nout,*) 'Approximate error estimates for right ', &
          'eigenvectors of (A,B)'
        Write (nout,99998)(eps*stnrm/dif(i),i=1,m)
      End If

99999 Format (' The 2-by-2 block (',I5,':',I5,') does not have a co', &
        'mplex eigenvalue')
99998 Format ((3X,1P,7E11.1))
    End Program f08ylfe
```

9.2 Program Data

```
F08YLF Example Program Data
  4                     :Value of N
  4.0  1.0  1.0  2.0
  0.0  3.0 -1.0  1.0
  0.0  1.0  3.0  1.0
  0.0  0.0  0.0  6.0    :End of matrix A
  2.0  1.0  1.0  3.0
  0.0  1.0  0.0  1.0
  0.0  0.0  1.0  1.0
  0.0  0.0  0.0  2.0    :End of matrix B
```

9.3 Program Results

```
 F08YLF Example Program Results

 S
       1.6E+00    1.7E+00    1.7E+00    1.4E+00

 DIF
       5.4E-01    1.5E-01    1.5E-01    1.2E-01

 Approximate error estimates for eigenvalues of (A,B)
       8.7E-16    7.8E-16    7.8E-16    9.9E-16

 Approximate error estimates for right eigenvectors of (A,B)
       2.5E-15    9.0E-15    9.0E-15    1.1E-14
```